



# PROC DATASETS

SELECTION COVERED IN THE OFFICE COLLECTION TRAINING SOLUTION

## COPYING A DATASET

```
(1.1) proc copy in=work out=demo datecopy;
      select ds / memtype=data;
run;
```

```
(1.2) proc datasets nolist;
      copy in=work out=demo datecopy;
      select ds / memtype=data;
run;
quit;
```

## MOVING A DATASET

```
(2.1) proc copy in=work out=demo datecopy move;
      select ds / memtype=data;
run;
```

```
(2.2) proc datasets nolist;
      copy in=work out=demo datecopy move;
      select ds / memtype=data;
run;
quit;
```

## APPENDING A DATASET

```
(3.1) proc append base=ds
      data=ds2;
run;
```

```
(3.2) proc datasets lib=work memtype=data nolist;
      append base=ds
      data=ds2;
run;
quit;
```

## RENAMING A DATASET

```
(4) proc datasets lib=work memtype=data nolist;
      change ds1old=ds1new
      ds2old=ds2new;
run;
quit;
```

## DISPLAYING DATASET METADATA

```
(5.1) ods select position;
proc contents data=ds varnum;
run;
```

```
(5.2) ods select position;
proc datasets nolist;
      contents data=ds varnum;
run;
quit;
```

```
(6.1) ods select variables;
proc contents data=ds;
run;
```

```
(6.2) ods select variables;
proc datasets nolist;
      contents data=ds;
run;
quit;
```

## ADDING A DATASET LABEL

```
(7.1) data ds (label='Dataset Label');
      set ds;
run;
```

```
(7.2) proc datasets lib=work memtype=data nolist;
      modify ds (label='Dataset Label');
run;
quit;
```

Be careful! When using a data step or proc sql, the dataset label is not copied over to the new dataset. The dataset label is kept when using proc copy or proc append.

## RENAMING VARIABLES

```
(8) proc datasets lib=work memtype=data nolist;
      modify ds;
      rename var1old=var1new
      var2old=var2new;
run;
quit;
```

## ADDING VARIABLE FORMAT AND LABEL

```
(9.1) proc datasets lib=work memtype=data nolist;
      modify ds;
      label var1new='Var. 1 Label'
      var2new='Var. 2 Label';
      format var1new var1fmt.
      var2new $var2fmt.;
run;
quit;
```

```
(9.2) proc datasets lib=work memtype=data nolist;
      modify ds;
      attrib var1new label='Var. 1 Label'
      format=var1fmt.
      var2new label='Var. 2 Label'
      format=$var2fmt.;
run;
quit;
```

## REMOVING VARIABLE FMT AND LABEL

```
(10.1) proc datasets lib=work memtype=data nolist;
      modify ds;
      label var1new=' '
      var2new=' ';
      format var1new var2new;
      *format _all_;
run;
quit;
```

```
(10.2) proc datasets lib=work memtype=data nolist;
      modify ds;
      attrib _all_ label=' ';
      attrib _all_ format=;
run;
quit;
```

## DELETING A SPECIFIC DATASET

```
(11.1) proc datasets lib=work memtype=data;
      delete ds;
run;
quit;
```

## DELETING ALL DATASETS

```
(11.2) proc datasets lib=work memtype=data kill;
run;
quit;
```

## DELETING ALL MEMBERS

```
(11.3) proc datasets lib=work kill; *memtype=all;
run;
quit;
```